Rules for integrands of the form $(c\,x)^m P_q[x] \left(a+b\,x^2\right)^p$

1: $\int x^m P_q\left[x^2\right] \left(a+b\,x^2\right)^p \, dx$ when $\frac{m-1}{2} \in \mathbb{Z}$

- **Derivation: Integration by substitution**
- **Basis: If $\frac{m-1}{2} \in \mathbb{Z}$, then $x^m F\left[x^2\right] == \frac{1}{2} \text{Subst}\left[x^{\frac{m-1}{2}} F[x],\, x,\, x^2\right] \partial_x x^2$**
- **Rule 1.1.2.y.1: If $\frac{m-1}{2} \in \mathbb{Z}$, then**

$$\int x^m P_q\left[x^2\right] \left(a+b\,x^2\right)^p \, dx \longrightarrow \frac{1}{2} \text{Subst}\left[\int x^{\frac{m-1}{2}} P_q[x] \left(a+b\,x\right)^p \, dx,\, x,\, x^2\right]$$

- **Program code:**

```
Int[x_^m_.*Pq_*(a_+b_.*x_^2)^p_.,x_Symbol] :=
  1/2*Subst[Int[x^((m-1)/2)*SubstFor[x^2,Pq,x]*(a+b*x)^p,x],x,x^2] /;
FreeQ[{a,b,p},x] && PolyQ[Pq,x^2] && IntegerQ[(m-1)/2]
```

2: $\int (c\,x)^m P_q[x] \left(a+b\,x^2\right)^p \, dx$ when $P_q[x,\,0] == 0$

- **Derivation: Algebraic simplification**
- **Rule 1.1.2.y.2: If $P_q[x,\,0] == 0$, then**

$$\int (c\,x)^m P_q[x] \left(a+b\,x^2\right)^p \, dx \longrightarrow \frac{1}{c} \int (c\,x)^{m+1} \text{PolynomialQuotient}\left[P_q[x],\, x,\, x\right] \left(a+b\,x^2\right)^p \, dx$$

- **Program code:**

```
Int[(c_.*x_)^m_.*Pq_*(a_+b_.*x_^2)^p_.,x_Symbol] :=
  1/c*Int[(c*x)^(m+1)*PolynomialQuotient[Pq,x,x]*(a+b*x^2)^p,x] /;
FreeQ[{a,b,c,m,p},x] && PolyQ[Pq,x] && EqQ[Coeff[Pq,x,0],0]
```

3: $\int (c\,x)^m \left(a+b\,x^2\right)^p \left(f+h\,x^2\right) dx$ when $a\,h\,(m+1)-b\,f\,(m+2\,p+3)=0\ \land\ m\neq -1$

- **Derivation: Special case of one step of the Ostrogradskiy-Hermite integration method**
- **Rule 1.1.2.y.3: If** $a\,h\,(m+1)-b\,f\,(m+2\,p+3)=0\ \land\ m\neq -1$**, then**

$$\int (c\,x)^m \left(a+b\,x^2\right)^p \left(f+h\,x^2\right) dx \longrightarrow \frac{f\,(c\,x)^{m+1}\left(a+b\,x^2\right)^{p+1}}{a\,c\,(m+1)}$$

- **Program code:**

```
Int[(c_.*x_)^m_.*P2_*(a_+b_.*x_^2)^p_.,x_Symbol] :=
  With[{f=Coeff[P2,x,0],g=Coeff[P2,x,1],h=Coeff[P2,x,2]},
  h*(c*x)^(m+1)*(a+b*x^2)^(p+1)/(b*c*(m+2*p+3)) /;
 EqQ[g,0] && EqQ[a*h*(m+1)-b*f*(m+2*p+3),0]] /;
FreeQ[{a,b,c,m,p},x] && PolyQ[P2,x,2] && NeQ[m,-1]
```

4: $\int (c\,x)^m P_q[x] \left(a+b\,x^2\right)^p dx$ when $p+2\in\mathbb{Z}^+$

Derivation: Algebraic expansion

- **Rule 1.1.2.y.4: If** $p+2\in\mathbb{Z}^+$**, then**

$$\int (c\,x)^m P_q[x] \left(a+b\,x^2\right)^p dx \longrightarrow \int \text{ExpandIntegrand}\left[(c\,x)^m P_q[x] \left(a+b\,x^2\right)^p,\ x\right] dx$$

- **Program code:**

```
Int[(c_.*x_)^m_.*Pq_*(a_+b_.*x_^2)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(c*x)^m*Pq*(a+b*x^2)^p,x],x] /;
FreeQ[{a,b,c,m},x] && PolyQ[Pq,x] && IGtQ[p,-2]
```

5: $\int x^m P_q\left[x^2\right] \left(a+b\,x^2\right)^p dx$ when $\frac{m}{2}\in\mathbb{Z}\ \bigwedge\ \frac{m+1}{2}+p\in\mathbb{Z}^-\ \bigwedge\ m+2\,q+2\,p+1<0$

- **Derivation: Algebraic expansion and binomial recurrence 3b**
- **Basis:** $\int x^m \left(a+b\,x^2\right)^p dx = \frac{x^{m+1}\left(a+b\,x^2\right)^{p+1}}{a\,(m+1)} - \frac{b\,(m+2\,(p+1)+1)}{a\,(m+1)}\int x^{m+2}\left(a+b\,x^2\right)^p dx$
- **Note: Interestingly this rule eleminates the constant term of** $P_q\left[x^2\right]$ **rather than the highest degree term.**
- **Rule 1.1.2.y.5: If** $\frac{m}{2}\in\mathbb{Z}\ \bigwedge\ \frac{m+1}{2}+p\in\mathbb{Z}^-\ \bigwedge\ m+2\,q+2\,p+1<0$**, let** $A\to P_q\left[x^2,\ 0\right]$ **and** $Q_{q-1}\left[x^2\right]\to \text{PolynomialQuotient}\left[P_q\left[x^2\right]-A,\ x^2,\ x\right]$**, then**

$$\int x^m P_q[x^2] \left(a+b x^2\right)^p dx \longrightarrow$$

$$A \int x^m \left(a+b x^2\right)^p dx + \int x^{m+2} Q_{q-1}[x^2] \left(a+b x^2\right)^p dx \longrightarrow$$

$$\frac{A x^{m+1} \left(a+b x^2\right)^{p+1}}{a (m+1)} + \frac{1}{a (m+1)} \int x^{m+2} \left(a+b x^2\right)^p \left(a (m+1) Q_{q-1}[x^2] - A b (m+2 (p+1)+1)\right) dx$$

Program code:

```
Int[x_^m_*Pq_*(a_+b_.*x_^2)^p_,x_Symbol] :=
  With[{A=Coeff[Pq,x,0],Q=PolynomialQuotient[Pq-Coeff[Pq,x,0],x^2,x]},
  A*x^(m+1)*(a+b*x^2)^(p+1)/(a*(m+1)) + 1/(a*(m+1))*Int[x^(m+2)*(a+b*x^2)^p*(a*(m+1)*Q-A*b*(m+2*(p+1)+1)),x]] /;
FreeQ[{a,b},x] && PolyQ[Pq,x^2] && IntegerQ[m/2] && ILtQ[(m+1)/2+p,0] && LtQ[m+Expon[Pq,x]+2*p+1,0]
```

6. $\int (c\,x)^m P_q[x]\,\left(a+b\,x^2\right)^p\,dx$ when $p<-1$

1: $\int (c\,x)^m P_q[x]\,\left(a+b\,x^2\right)^p\,dx$ when $p<-1\land m>0$

Derivation: Algebraic expansion and quadratic recurrence 2a

Rule 1.1.2.y.6.1: If $p<-1\land m>0$,

let $Q_{q-2}[x]\to\text{PolynomialQuotient}\left[P_q[x],\ a+b\,x^2,\ x\right]$ and $f+g\,x\to\text{PolynomialRemainder}\left[P_q[x],\ a+b\,x^2,\ x\right]$, then

$$\int (c\,x)^m P_q[x]\,\left(a+b\,x^2\right)^p\,dx\ \longrightarrow$$

$$\int (c\,x)^m\,(f+g\,x)\,\left(a+b\,x^2\right)^p\,dx+\int (c\,x)^{m-1}\,(c\,x)\,Q_{q-2}[x]\,\left(a+b\,x^2\right)^{p+1}\,dx\ \longrightarrow$$

$$\frac{(c\,x)^m\,\left(a+b\,x^2\right)^{p+1}\,(a\,g-b\,f\,x)}{2\,a\,b\,(p+1)}+\frac{c}{2\,a\,b\,(p+1)}\int (c\,x)^{m-1}\,\left(a+b\,x^2\right)^{p+1}\,\left(2\,a\,b\,(p+1)\,x\,Q_{-2+q}[x]-a\,g\,m+b\,f\,(m+2\,p+3)\,x\right)\,dx$$

Program code:

```
Int[(c_.*x_)^m_.*Pq_*(a_+b_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[Pq,a+b*x^2,x],
        f=Coeff[PolynomialRemainder[Pq,a+b*x^2,x],x,0],
        g=Coeff[PolynomialRemainder[Pq,a+b*x^2,x],x,1]},
  (c*x)^m*(a+b*x^2)^(p+1)*(a*g-b*f*x)/(2*a*b*(p+1)) +
  c/(2*a*b*(p+1))*Int[(c*x)^(m-1)*(a+b*x^2)^(p+1)*ExpandToSum[2*a*b*(p+1)*x*Q-a*g*m+b*f*(m+2*p+3)*x,x],x]] /;
FreeQ[{a,b,c},x] && PolyQ[Pq,x] && LtQ[p,-1] && GtQ[m,0]
```

2. $\int (c\,x)^m P_q[x]\,\left(a+b\,x^2\right)^p\,dx$ when $p<-1\land m\not>0$

1: $\int (c\,x)^m P_q[x]\,\left(a+b\,x^2\right)^p\,dx$ when $p<-1\land m\in\mathbb{Z}^-$

Derivation: Algebraic expansion and trinomial recurrence 2b

Rule 1.1.2.y.6.2.1: If $p<-1\land m\in\mathbb{Z}^-$,

let $Q_{m+q-2}[x]\to\text{PolynomialQuotient}\left[(c\,x)^m P_q[x],\ a+b\,x^2,\ x\right]$ and $f+g\,x\to\text{PolynomialRemainder}\left[(c\,x)^m P_q[x],\ a+b\,x^2,\ x\right]$, then

$$\int (c\,x)^m P_q[x]\,\left(a+b\,x^2\right)^p\,dx\ \longrightarrow$$

$$\int (f + g\,x)\,\left(a + b\,x^2\right)^p\,dx + \int Q_{m+q-2}[x]\,\left(a + b\,x^2\right)^{p+1}\,dx \longrightarrow$$

$$\frac{(a\,g - b\,f\,x)\,\left(a + b\,x^2\right)^{p+1}}{2\,a\,b\,(p+1)} + \frac{1}{2\,a\,(p+1)}\int (c\,x)^m\,\left(a + b\,x^2\right)^{p+1}\,\left(2\,a\,(p+1)\,(c\,x)^{-m}\,Q_{m+q-2}[x] + f\,(2\,p+3)\,(c\,x)^{-m}\right)dx$$

Program code:

```
Int[(c_.*x_)^m_.*Pq_*(a_+b_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[(c*x)^m*Pq,a+b*x^2,x],
        f=Coeff[PolynomialRemainder[(c*x)^m*Pq,a+b*x^2,x],x,0],
        g=Coeff[PolynomialRemainder[(c*x)^m*Pq,a+b*x^2,x],x,1]},
  (a*g-b*f*x)*(a+b*x^2)^(p+1)/(2*a*b*(p+1)) +
  1/(2*a*(p+1))*Int[(c*x)^m*(a+b*x^2)^(p+1)*ExpandToSum[2*a*(p+1)*(c*x)^(-m)*Q+f*(2*p+3)*(c*x)^(-m),x],x]] /;
FreeQ[{a,b,c},x] && PolyQ[Pq,x] && LtQ[p,-1] && ILtQ[m,0]
```

2: $\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^2\right)^p\,dx$ when $p<-1\,\wedge\,m\not>0$

Derivation: Algebraic expansion and quadratic recurrence 2b

Rule 1.1.2.y.6.2.2: If $p<-1\,\wedge\,m\not>0$,
let $Q_{q-2}[x]\to\text{PolynomialQuotient}\left[P_q[x],\,a+b\,x^2,\,x\right]$ and $f+g\,x\to\text{PolynomialRemainder}\left[P_q[x],\,a+b\,x^2,\,x\right]$, then

$$\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^2\right)^p\,dx\,\longrightarrow$$

$$\int (c\,x)^m\,(f+g\,x)\,\left(a+b\,x^2\right)^p\,dx+\int (c\,x)^m\,Q_{q-2}[x]\,\left(a+b\,x^2\right)^{p+1}\,dx\,\longrightarrow$$

$$-\frac{(c\,x)^{m+1}\,(f+g\,x)\,\left(a+b\,x^2\right)^{p+1}}{2\,a\,c\,(p+1)}+\frac{1}{2\,a\,(p+1)}\int (c\,x)^m\,\left(a+b\,x^2\right)^{p+1}\,\left(2\,a\,(p+1)\,Q_{q-2}[x]+f\,(m+2\,p+3)+g\,(m+2\,p+4)\,x\right)\,dx$$

Program code:

```
Int[(c_.*x_)^m_.*Pq_*(a_+b_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[Pq,a+b*x^2,x],
        f=Coeff[PolynomialRemainder[Pq,a+b*x^2,x],x,0],
        g=Coeff[PolynomialRemainder[Pq,a+b*x^2,x],x,1]},
  -(c*x)^(m+1)*(f+g*x)*(a+b*x^2)^(p+1)/(2*a*c*(p+1)) +
  1/(2*a*(p+1))*Int[(c*x)^m*(a+b*x^2)^(p+1)*ExpandToSum[2*a*(p+1)*Q+f*(m+2*p+3)+g*(m+2*p+4)*x,x],x]] /;
FreeQ[{a,b,c,m},x] && PolyQ[Pq,x] && LtQ[p,-1] && Not[GtQ[m,0]]
```

7: $\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^2\right)^p\,dx$ when $m<-1$

Derivation: Algebraic expansion and quadratic recurrence 3b

Note: If $q==1$**, no need to reduce integrand since** $\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^2\right)^p\,dx$ **can be expressed as a two term sum of hyperbolic functions.**

Rule 1.1.2.y.7: If $m<-1$,
let $Q_{q-1}[x]\to\text{PolynomialQuotient}[P_q[x],\,c\,x,\,x]$ and $R\to\text{PolynomialRemainder}[P_q[x],\,c\,x,\,x]$, then

$$\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^2\right)^p\,dx\,\longrightarrow$$

$$\int (c\,x)^{m+1}\,Q_{q-1}[x]\,\left(a+b\,x^2\right)^p\,dx+R\int (c\,x)^m\,\left(a+b\,x^2\right)^p\,dx\,\longrightarrow$$

$$\frac{R\,(c\,x)^{m+1}\,\left(a+b\,x^2\right)^{p+1}}{a\,c\,(m+1)}+\frac{1}{a\,c\,(m+1)}\int (c\,x)^{m+1}\,\left(a+b\,x^2\right)^p\,\left(a\,c\,(m+1)\,Q_{q-1}[x]-b\,R\,(m+2\,p+3)\,x\right)\,dx$$

Program code:

```
Int[(c_.*x_)^m_*Pq_*(a_+b_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[Pq,c*x,x], R=PolynomialRemainder[Pq,c*x,x]},
  R*(c*x)^(m+1)*(a+b*x^2)^(p+1)/(a*c*(m+1)) +
  1/(a*c*(m+1))*Int[(c*x)^(m+1)*(a+b*x^2)^p*ExpandToSum[a*c*(m+1)*Q-b*R*(m+2*p+3)*x,x],x]] /;
FreeQ[{a,b,c,p},x] && PolyQ[Pq,x] && LtQ[m,-1] && (IntegerQ[2*p] || NeQ[Expon[Pq,x],1])
```

8: $\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^2\right)^p\,dx$ when $m+q+2\,p+1==0$

Derivation: Algebraic expansion

- **Basis:** $(c\,x)^m\,P_q[x]==\frac{P_q[x,q]\,(c\,x)^{m+q}}{c^q}+\frac{(c\,x)^m\,\left(c^q\,P_q[x]-P_q[x,q]\,(c\,x)^q\right)}{c^q}$

Rule 1.1.2.y.8: If $m+q+2\,p+1==0$**, then**

$$\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^2\right)^p\,dx\,\longrightarrow$$

$$\frac{P_q[x,\,q]}{c^q}\int (c\,x)^{m+q}\,\left(a+b\,x^2\right)^p\,dx+\frac{1}{c^q}\int (c\,x)^m\,\left(a+b\,x^2\right)^p\,\left(c^q\,P_q[x]-P_q[x,\,q]\,(c\,x)^q\right)\,dx$$

- **Program code:**

```
Int[(c_.*x_)^m_.*Pq_*(a_+b_.*x_^2)^p_.,x_Symbol] :=
  With[{q=Expon[Pq,x]},
  Coeff[Pq,x,q]/c^q*Int[(c*x)^(m+q)*(a+b*x^2)^p,x] +
  1/c^q*Int[(c*x)^m*(a+b*x^2)^p*ExpandToSum[c^q*Pq-Coeff[Pq,x,q]*(c*x)^q,x],x] /;
 EqQ[q,1] || EqQ[m+q+2*p+1,0]] /;
FreeQ[{a,b,c,m,p},x] && PolyQ[Pq,x] && Not[IGtQ[m,0] && ILtQ[p+1/2,0]]
```

9: $\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^2\right)^p\,dx$ when $q>1\,\bigwedge\,m+q+2\,p+1\neq 0\,\bigwedge\,\left(m\notin\mathbb{Z}^+\bigvee\,p+\frac{1}{2}+1\in\mathbb{Z}^+\right)$

- **Derivation: Algebraic expansion and quadratic recurrence 3a with** $A=d$**,** $B=e$ **and** $m=m-1$

- **Rule 1.1.2.y.9: If** $q>1\,\bigwedge\,m+q+2\,p+1\neq 0\,\bigwedge\,\left(m\notin\mathbb{Z}^+\bigvee\,p+\frac{1}{2}+1\in\mathbb{Z}^+\right)$**, let** $f\to P_q[x,\,q]$**, then**

$$\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^2\right)^p\,dx\,\longrightarrow$$

$$\int (c\,x)^m \left(P_q[x] - \frac{f}{c^q}\,(c\,x)^q\right)\left(a + b\,x^2\right)^p\,dx + \frac{f}{c^q}\int (c\,x)^{m+q}\left(a + b\,x^2\right)^p\,dx \longrightarrow$$

$$\frac{f\,(c\,x)^{m+q-1}\left(a + b\,x^2\right)^{p+1}}{b\,c^{q-1}\,(m + q + 2\,p + 1)} +$$

$$\frac{1}{b\,(m + q + 2\,p + 1)}\int (c\,x)^m\left(a + b\,x^2\right)^p\left(b\,(m + q + 2\,p + 1)\,P_q[x] - b\,f\,(m + q + 2\,p + 1)\,x^q - a\,f\,(m + q - 1)\,x^{q-2}\right)dx$$

- **Program code:**

```
Int[(c_.*x_)^m_.*Pq_*(a_+b_.*x_^2)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x],f=Coeff[Pq,x,Expon[Pq,x]]},
  f*(c*x)^(m+q-1)*(a+b*x^2)^(p+1)/(b*c^(q-1)*(m+q+2*p+1)) +
  1/(b*(m+q+2*p+1))*Int[(c*x)^m*(a+b*x^2)^p*ExpandToSum[b*(m+q+2*p+1)*Pq-b*f*(m+q+2*p+1)*x^q-a*f*(m+q-1)*x^(q-2),x],x] /;
 GtQ[q,1] && NeQ[m+q+2*p+1,0]] /;
FreeQ[{a,b,c,m,p},x] && PolyQ[Pq,x] && (Not[IGtQ[m,0]] || IGtQ[p+1/2,-1])
```